
Package: pod2pdf
Version: 0.42-4
Architecture: all
Maintainer: Guo Yixuan <culu.gyx@gmail.com>
Installed-Size: 94
Depends: perl, perl-modules, libpdf-api2-perl (>= 0.6), libgetopt-argvfile-perl
Recommends: libimage-size-perl, libfile-type-perl
Provides: libpod2pdf-perl
Section: perl
Priority: optional
Homepage: http://search.cpan.org/dist/pod2pdf/
Description: Plain Old Documentation to Portable Document Format converter
 POD allows for the documentation of Perl code. But with
 a few tricks, e.g., when disguised as comments, it is
 applicable to many other languages, too. It resembles
 HTML in the way that it allows for the structured representation
 of text. Hence, it looks rather good when converted for
 printers, HTML or man pages.
 .
 This package represents more than a mere alternative
 to pod2man|groff -man to achieve a coversion from POD
 to PostScript and from there via ps2pdf to PDF.
 Its major advantage lies in the inclusion of images
 for the documentation. This renders POD suitable for
 many smallish projects as a lingua franca for their
 documentation - associated with the source and separate
 documents - so texts (and graphics) can be shared more
 easily between documenters and programmers.











data.tar.gz
data.tar




./usr/share/perl5/App/pod2pdf.pm


# App::pod2pdf
#
# Copyright (C) 2007 Jon Allen <jj@jonallen.info>
#
# This software is licensed under the terms of the Artistic
# License version 2.0.
#
# For full license details, please read the file 'artistic-2_0.txt' 
# included with this distribution, or see
# http://www.perlfoundation.org/legal/licenses/artistic-2_0.html

package App::pod2pdf;

use strict;
use warnings;
use Carp;
use List::Util qw/max min/;
use PDF::API2;
use Pod::Escapes qw/e2char/;
use Pod::Parser;
use Pod::ParseLink;

use constant TRUE  => 1;
use constant FALSE => 0;

BEGIN {
  our @ISA     = qw/Pod::Parser/;
  our $VERSION = '0.42';
}


#-----------------------------------------------------------------------
#-----------------------------------------------------------------------

sub new {
  my $invocant = shift;
  my $class    = ref($invocant) || $invocant;

  my %user_options    = @_;
  my %default_options = (
    header        => TRUE,         # Include header on all pages
    footer        => TRUE,         # Include footer on all pages
    page_width    => 595,          # A4
    page_height   => 842,          # A4
    left_margin   => $user_options{margins} || 48,  # 0.75"
    right_margin  => $user_options{margins} || 48,  # 0.75"
    top_margin    => $user_options{margins} || 60,  # 
    bottom_margin => $user_options{margins} || 60,  # 
    font_face     => 'Helvetica',  # Sans-Serif text
    font_size     => 10,           # Text size = 10 points
    icon_scale    => 0.25,         # Icon scaling (%age)
    );
  
  my $self = $class->SUPER::new(%default_options,%user_options);
  
  $self->create_pdf;
  return $self;
}


#-----------------------------------------------------------------------

sub command {
  my ($self, $command, $paragraph, $line_num) = @_;
  my $expansion = $self->interpolate($paragraph, $line_num);  

  COMMAND: {
    if ($command eq 'ff') {
     $self->formfeed if ($self->print_flag);
    }
    if ($command =~ /^head[1234]$/) {
      $self->indent(0);
      $self->set_style('default');
      $self->newline;
      my $default_space = $self->{line_spacing};
      $self->set_style($command);
      my $heading_space = $self->{line_spacing};
      # Checks to see if there is space for a content line after
      # the heading - if not then starts a new page
      if ( ($self->{y_position} - $heading_space - $default_space - $self->{spacer}) < 
           ($self->{bottom_margin} + $self->{footer_height}) ) {
        $self->formfeed;
      } else {
        $self->{y_position} -= ($heading_space - $default_space);
      }
      # add an outline at the heading's position
      $self->generate_outline($expansion, $command, $self->{page_number}, $self->{x_position}, $self->{y_position} + $heading_space) if $self->{outlines};
      $self->print_text_with_style($expansion,$command);
      $self->spacer;
      $self->indent(48);
    }
    if ($command eq 'over') {
      my $indentlevel = $expansion || 4;
      $self->set_style;
      $self->push_indent($indentlevel * $self->em);
      $self->reset_item_textblock_flag;
    }
    if ($command eq 'back') {
      $self->pop_indent;
      $self->spacer;
    }
    if ($command eq 'item') {
      $self->spacer if ($self->item_textblock_flag);
      $self->reset_item_textblock_flag;
      if ($expansion =~ '^\s*\*?\s*$') {
        # First check to see if there is space for any text
        if ($self->{y_position} - $self->{line_spacing} < ($self->{bottom_margin} + $self->{footer_height})) {
          $self->formfeed;
        } 
        my $indent = $self->pop_indent;
        $self->bullet($indent);
        $self->push_indent($indent);
      } elsif ($expansion =~ '^\s*(\d+\.?)\s*$') {
        # First check to see if there is space for any text
        if ($self->{y_position} - $self->{line_spacing} < ($self->{bottom_margin} + $self->{footer_height})) {
          $self->formfeed;
        } 
        my $indent = $self->pop_indent;
        $self->{y_position} -= $self->{line_spacing};
        $self->print_text_with_style($1,'default');
        $self->push_indent($indent);
        $self->{y_position} += $self->{line_spacing};
      } else {
        my $indent = $self->pop_indent;
        $self->set_style;
        $self->newline;
        $self->parse_text({-expand_ptree => 'print_tree'},$paragraph,$line_num);
        $self->spacer;    
        $self->push_indent($indent);
      }
    }
  }
}


#-----------------------------------------------------------------------

sub verbatim {
  my ($self, $paragraph, $line_num) = @_;
  if ($paragraph =~ /^[ \t]/) {
    $self->set_style('verbatim');
    $self->reset_space_flag;
    $self->set_item_textblock_flag;
    foreach my $line (split /\n/,$paragraph) {
      # todo: expand tabs
      if ($line =~ /\S/) {
        $self->newline;
        $self->print_text_with_style($line,'verbatim');
        $self->reset_space_flag;
      }
    }
    $self->newline;
    $self->spacer unless ($self->over);
  }
}


#-----------------------------------------------------------------------

sub textblock {
  my ($self, $text, $line_num) = @_;
  if ($text =~ /\S/) {  # ignore blank paragraphs
    $self->set_item_textblock_flag;
    $self->reset_space_flag;
    $self->set_style;
    $self->newline;
    $self->parse_text({-expand_ptree => 'print_tree'},$text,$line_num);
    $self->spacer;
    $self->spacer unless ($self->over);
  }
}


#-----------------------------------------------------------------------

sub interior_sequence {
  my ($self,$command,$text) = @_;
  #
  # need to check content of $text, i.e.
  # is there a nested formatting command?
  #
  # also this doesn't handle the L<> formatting
  # command, check with perlpodspec if this is
  # allowed in =head blocks
  #
  COMMAND: {
    if ($command eq 'X') {
      # no-op
      last COMMAND;
    }
    if ($command eq 'Z') {
      # no-op
      last COMMAND;
    }
    if ($command eq 'E') {
      return e2char($text);
    }
    DEFAULT: {
      return $text;
    }
  }
}


#-----------------------------------------------------------------------

sub print_tree {
  my $self = shift;
  my $tree = shift;
  NODE: foreach my $node ($tree->children) {
    if (ref $node) {
      COMMAND: { 
        my $command = $node->cmd_name;
        if ($command eq 'L') {
          #warn("Found link: ".$node->raw_text."\n");
          my $left_delimiter  = $node->left_delimiter;
          my $right_delimiter = $node->right_delimiter;
          (my $link_text = $node->raw_text) =~ s/L$left_delimiter\s*(.*?)\s*$right_delimiter$/$1/s;
          my ($text, $inferred, $name, $section, $type) = parselink($link_text);
	  $text     =~ s/^"(.*?)"$/$1/ if ($text);
	  $inferred =~ s/^"(.*?)"$/$1/ if ($inferred);
	  $name     =~ s/^"(.*?)"$/$1/ if ($name);
          $self->push_format('I');
          $self->parse_text({-expand_ptree => 'print_tree'},($text || $inferred || $name));
          $self->pop_format;
          last COMMAND;
        }
        if ($command eq 'O') {
          my $left_delimiter  = $node->left_delimiter;
          my $right_delimiter = $node->right_delimiter;
          (my $object_text = $node->raw_text) =~ s/O$left_delimiter\s*(.*?)\s*$right_delimiter$/$1/;
          my ($object_title,$object_location) = parseobject($object_text);
          if ($object_location =~ /\A\W+:[^:\s]\S*\z/) {
            # URL - cannot load (yet!)
            $self->warnonce('HTTP object loading not supported');
            $self->print_text_with_style($object_location,'I');
          } elsif (-e $object_location) {
            # Found file
            if ($self->images) {
              my $mime_type = File::Type->new->mime_type($object_location);
              if ($mime_type =~ /^image/) {
                unless ($self->insert_image($object_location)) {
                  $self->print_text_with_style($object_location,'I');                
                }
              } else {
                $self->print_text_with_style($object_location,'I');
              }
            } else {
              $self->print_text_with_style($object_location,'I');
            }
          } else {
            # Non-existant file
            $self->warnonce("Object not found: $object_location");
            $self->print_text_with_style("Object not found: $object_location",'I');
          }
          last COMMAND;
        }
        if ($command eq 'X') {
          # no-op
        }
        DEFAULT: {
          $self->push_format($node->cmd_name);
          $self->print_tree($node->parse_tree);
          $self->pop_format;
        }
      }
    } else {
      FORMAT: {
        $_ = $self->format;
        if (/X/) {
          # no-op
          last FORMAT;
        }
        if (/Z/) {
          # no-op
          last FORMAT;
        }
        if (/E/) {
          $node = e2char($node);
        }
        if (/BC.*I/) {
          $self->print_text_with_style($node,'BCI');
          last FORMAT;
        }
        if (/C.*I/) {
          $self->print_text_with_style($node,'CI');
          last FORMAT;
        }
        if (/B.*I/) {
          $self->print_text_with_style($node,'BI');        
          last FORMAT;
        }
        if (/BC/) {
          $self->print_text_with_style($node,'BC');        
          last FORMAT;
        }
        if (/B/) {
          $self->print_text_with_style($node,'B');        
          last FORMAT;
        }
        if (/C/) {
          $self->print_text_with_style($node,'C');        
          last FORMAT;
        }
        if (/I/) {
          $self->print_text_with_style($node,'I');        
          last FORMAT;
        }
        DEFAULT: {
          #warn "Line 414: $_\n";
          $self->print_text_with_style($node,'default');
          last FORMAT;
        }
      }
    }
  }
}


#-----------------------------------------------------------------------

sub insert_image {
  my $self     = shift;
  my $filename = shift;
  
  if ($self->images) {
    if (-e $filename) {
      my $image;
      my $type = File::Type->new->checktype_filename($filename);
      SWITCH: {
        if ($type eq 'image/jpeg')  {$image = $self->{pdf}->image_jpeg($filename); last}
        if ($type eq 'image/tiff')  {$image = $self->{pdf}->image_tiff($filename); last}
        if ($type eq 'image/gif')   {$image = $self->{pdf}->image_gif($filename);  last}
        if ($type eq 'image/x-png') {$image = $self->{pdf}->image_png($filename);  last}
        if ($type eq 'image/x-pnm') {$image = $self->{pdf}->image_pnm($filename);  last}
        $self->warnonce("[Warning] Unknown image format '$type' for image '$filename'");
        return FALSE;
      }
      
      unless ($image) {
        $self->warnonce("[Warning] Cannot load image file '$filename'");
        return FALSE;
      }

      my ($width,$height)  = imgsize($filename);
      my $available_width  = $self->{page_width} - $self->{right_margin} - $self->{x_position};
      my $scale_default    = 0.5;
      my $scale_min        = 0.4;
      my $scale            = min($available_width / $width, $scale_default);
      my $height_in_points = $height * $scale;

      if ($self->{y_position} < ($self->{bottom_margin} + $self->{footer_height} + $height_in_points + ($self->{line_spacing} / 2))) {
        my $available_height = $self->{y_position} - $self->{bottom_margin} - $self->{footer_height} - $self->{line_spacing};
        if ($available_height / $height > $scale_min) {
          $scale = $available_height / $height;
          $height_in_points = $height * $scale;
        } else {
          $self->formfeed;
          $self->set_print_flag;
        }
      }
  
      $self->{y_position} -= $height_in_points;
      $self->{y_position} += ($self->{line_spacing} / 2);
      $self->{gfx}         = $self->{page}->gfx unless (exists $self->{gfx});  
      $self->{gfx}->image($image,$self->{x_position},$self->{y_position},$scale);
      return TRUE;
    } else {
      $self->warnonce("Image '$filename' does not exist");
      return FALSE;
    }
  }
}


#-----------------------------------------------------------------------

sub images {
  my $self = shift;
  
  unless ($self->{image_modules_check}) {
    # Check if image modules are installed
    eval "use File::Type;use Image::Size;";
    if ($@) {
      $self->warnonce('Cannot use images, modules Image::Size and/or File::Type not installed');
    } else {      
      $self->{image_modules_loaded} = TRUE;
    }
    $self->{image_modules_check}  = TRUE;
  }
  
  return $self->{image_modules_loaded};  
}


#-----------------------------------------------------------------------

sub warnonce {
  my $self    = shift;
  my $warning = shift;
  
  unless ($self->{issued_warnings}->{$warning}) {
    warn("[Warning] $warning\n");
    $self->{issued_warnings}->{$warning} = TRUE;
  }
}


#-----------------------------------------------------------------------

sub parseobject {
  # Parses the O<...> formatting code as specified in perlpodextensions
  my $object_text = shift;
  if ($object_text =~ /(.*?)\|(.*)/) {
    return ($1,$2);
  } else {
    return (undef,$object_text);
  }
}


#-----------------------------------------------------------------------

sub create_pdf {
  my $self    = shift;
  my $class   = ref $self;
  my $version = $self->VERSION();
      
  # Define styles
  #
  # Future enhancement: move the style definitions into a separate
  # module (e.g. Pod::Pdf::Styles) which can be subclassed to allow
  # non-core fonts to be used.
  #
  $self->{stylist} = {
    'header'      => {font=>'Helvetica-Bold',        size=>10       },
    'footer'      => {font=>'Helvetica-Bold',        size=>10       },
    'head1'       => {font=>'Helvetica-Bold',        size=>12       },
    'head2'       => {font=>'Helvetica-Bold',        size=>11       },
    'head3'       => {font=>'Helvetica-Bold',        size=>10       },
    'head4'       => {font=>'Helvetica',             size=>10       },
    'verbatim'    => {font=>'Courier',               verbatim=>TRUE },
    'B'           => {font=>'Helvetica-Bold'                        },
    'BC'          => {font=>'Courier-Bold',          verbatim=>TRUE },
    'BI'          => {font=>'Helvetica-BoldOblique'                 },
    'BCI'         => {font=>'Courier-BoldOblique',   verbatim=>TRUE },
    'C'           => {font=>'Courier',               verbatim=>TRUE },
    'CI'          => {font=>'Courier-Oblique',       verbatim=>TRUE },
    'I'           => {font=>'Helvetica-Oblique'                     },
  };
  
  # Set up first page
  PAGE_SIZE: {
    if ($self->{page_size}) {
      eval "use Paper::Specs 0.10 units=>'pt';";
      if ($@) {
        $self->warnonce("Cannot use '--page-size' option, module Paper::Specs (v0.10) not installed");
      } else {
        if (my $form = Paper::Specs->find(code=>$self->{page_size}, brand=>'standard')) {
          $self->{page_width}  = int($form->sheet_width + 0.5);
          $self->{page_height} = int($form->sheet_height + 0.5);
        } else {
          $self->warnonce("Unknown page size '".$self->{page_size}."'");
        } 
      }
    }
  }
  
  PAGE_ORIENTATION: {
    if ($self->{page_orientation}) {
      if (lc $self->{page_orientation} eq 'landscape') {
        ($self->{page_width},$self->{page_height}) = (
          max($self->{page_width},$self->{page_height}),
          min($self->{page_width},$self->{page_height}) 
        );
        last PAGE_ORIENTATION;
      }
      if (lc $self->{page_orientation} eq 'portrait') {
        ($self->{page_width},$self->{page_height}) = (
          min($self->{page_width},$self->{page_height}),
          max($self->{page_width},$self->{page_height}) 
        ); 
        last PAGE_ORIENTATION; 
      }
      $self->warnonce("Unknown page orientation '".$self->{page_orientation}."', must be 'portrait' or 'landscape'");
    }  
  }
  
  $self->{page_number}   = 0;
  $self->{line_spacing}  = $self->{font_size}+2 unless ($self->{line_spacing});
  $self->{x_position}    = $self->{left_margin};
  $self->{y_position}    = $self->{page_height} - $self->{top_margin};
  $self->{indent}        = 0;
  $self->{pdf}           = PDF::API2->new;
  $self->{pdf}->info('Producer'=>"$class version $version");
  $self->{pdf}->mediabox($self->{page_width},$self->{page_height});

  # push the root Outlines object into outline_stack
  push @{$self->{outline_stack}}, $self->{pdf}->outlines() if $self->{outlines};
  
  if ($self->{icon} && $self->images) {
      if (-e $self->{icon}) {
        my $type = File::Type->new->checktype_filename($self->{icon});
        SWITCH: {
          if ($type eq 'image/jpeg')  {$self->{icon_img} = $self->{pdf}->image_jpeg($self->{icon}); last}
          if ($type eq 'image/tiff')  {$self->{icon_img} = $self->{pdf}->image_tiff($self->{icon}); last}
          if ($type eq 'image/gif')   {$self->{icon_img} = $self->{pdf}->image_gif($self->{icon});  last}
          if ($type eq 'image/x-png') {$self->{icon_img} = $self->{pdf}->image_png($self->{icon});  last}
          if ($type eq 'image/x-pnm') {$self->{icon_img} = $self->{pdf}->image_pnm($self->{icon});  last}
          warn "[Warning] Unknown image format '$type' for icon ".$self->{icon}."\n";
        }
        if ($self->{icon_img}) {
          ($self->{icon_width},$self->{icon_height}) = imgsize($self->{icon});  
        }    
      } else {
        warn("[Warning] Cannot open icon file: ".$self->{icon}."\n");
      }
  }  

  $self->formfeed;
  $self->set_style;
  $self->{indent}        = 0;
  $self->{over}          = 0;
  $self->{spacer}        = 4;  # default spacing between paragraphs
}


#-----------------------------------------------------------------------
#-----------------------------------------------------------------------

# Item_textblock_flag methods
#
# This flag is used to control line spacing within =over sections. The
# flag is cleared after each =item command and set whenever a textblock
# is printed.
#
# At the start of processing an =item command, an extra half line space 
# (4 points) is inserted if the textblock flag is set. Because half
# spacing is the default in =over sections, this extra space between
# individual =items acts to visually group the =item paragraphs as a
# single element.

#-----------------------------------------------------------------------

sub item_textblock_flag {
  my $self = shift;
  return $self->{item_textblock_flag}->{$self->over} || 0;
}


#-----------------------------------------------------------------------

sub set_item_textblock_flag {
  my $self = shift;
  $self->{item_textblock_flag}->{$self->over} = TRUE;
}


#-----------------------------------------------------------------------

sub reset_item_textblock_flag {
  my $self = shift;
  $self->{item_textblock_flag}->{$self->over} = FALSE;
}


#-----------------------------------------------------------------------
#-----------------------------------------------------------------------

# Print_flag methods
#
# The Print flag is used to prevent blank lines from appearing at the 
# start of a page, which can happen if a verbatim block or =over list 
# crosses a page break.
#
# When a new page is started, the print flag is reset. In this state
# any calls to newline() or spacer() will have no effect. Whenever any
# text is printed, the print flag will be set, then newlines will
# operate nomally.

#-----------------------------------------------------------------------

sub print_flag {
  my $self = shift;
  return $self->{print_flag} || 0;
}


#-----------------------------------------------------------------------

sub set_print_flag {
  my $self = shift;
  $self->{print_flag} = TRUE;
}


#-----------------------------------------------------------------------

sub reset_print_flag {
  my $self = shift;
  $self->{print_flag} = FALSE;
}


#-----------------------------------------------------------------------
#-----------------------------------------------------------------------

# Space_flag methods
#
# The space flag is used to prevent the display of whitespace characters
# at the end of a paragraph. If these characters are not suppressed,
# then occasionally they will wrap onto the next line, causing unsightly 
# spaces in the finished document.
#
# Each string presented to the print_text_with_style() method is checked
# for trailling whitespace. If so, the space_flag is set. At the next
# call to print_text_with_style(), an extra space character is printed
# if the space_flag is set. The space_flag is cleared either when the
# spacer() method is called (to mark the 'real' end of a text block), or
# after the flag has caused a new space to be inserted.

#-----------------------------------------------------------------------

sub space_flag {
  my $self = shift;
  return $self->{space_flag} || 0;
}


#-----------------------------------------------------------------------

sub set_space_flag {
  my $self = shift;
  $self->{space_flag} = TRUE;
}


#-----------------------------------------------------------------------

sub reset_space_flag {
  my $self = shift;
  $self->{space_flag} = FALSE;
}


#-----------------------------------------------------------------------

sub flag {
  my $self = shift;
  my $flag = shift or return FALSE;
  return $self->{flags}->{$flag} || FALSE;
}


#-----------------------------------------------------------------------

sub set_flag {
  my $self = shift;
  my $flag = shift or return FALSE;
  $self->{flags}->{$flag} = TRUE;
}


#-----------------------------------------------------------------------

sub clear_flag {
  my $self = shift;
  my $flag = shift or return FALSE;
  $self->{flags}->{$flag} = FALSE;
}

#-----------------------------------------------------------------------
#-----------------------------------------------------------------------

# Text indent methods

#-----------------------------------------------------------------------

sub indent {
  # Sets the current indent (measured in points)
  my $self            = shift;
  $self->{indent}     = shift;
  $self->{x_position} = $self->{left_margin} + $self->{indent};
}


#-----------------------------------------------------------------------

sub over {
  # Returns the current number of nested =over blocks
  my $self = shift;
  return $self->{over};
}


#-----------------------------------------------------------------------

sub em {
  # Returns the width (in points) of an 'm' character, used by =over X
  # to decide how much to indent by
  my $self = shift;
  return $self->{mspace};
}


#-----------------------------------------------------------------------

sub push_indent {
  my $self   = shift;
  my $indent = shift;
  push @{$self->{indent_list}},$indent;
  $self->indent($self->{indent} + $indent);
  $self->{over}++;
}


#-----------------------------------------------------------------------

sub pop_indent {
  my $self = shift;
  $self->{over}--;
  if (@{$self->{indent_list}}) {
    my $indent = pop @{$self->{indent_list}};
    $self->indent($self->{indent} - $indent);
    return $indent;
  } else {
    return 0;
  }
}


#-----------------------------------------------------------------------
#-----------------------------------------------------------------------

# Text format methods
#
# During parsing, as each Pod::InteriorSequence object is encountered
# the formatting code (B, I, etc) is pushed onto a stack. When the
# parser gets to the individual text elements, the format() method will
# return the complete set of codes which need to be applied to the text.

#-----------------------------------------------------------------------

sub push_format {
  my $self   = shift;
  my $format = shift;
  push @{$self->{format}},$format;
}


#-----------------------------------------------------------------------

sub pop_format {
  my $self = shift;
  return pop @{$self->{format}} if (@{$self->{format}});
}


#-----------------------------------------------------------------------

sub format {
  # Returns the current text format as a scalar, e.g. 'BEI' for Bold
  # Italic with Escapes to be processed. Formatting codes are listed in
  # alphabetical order with duplicates removed.
  my $self = shift;
  my %format;
  foreach (@{$self->{format}}) {
    # Treat F<> as a synonym for I<> (renders filenames in italic)
    tr/F/I/;
    $format{$_}++;
  }
  return join '',sort keys %format;
}


#-----------------------------------------------------------------------
#-----------------------------------------------------------------------

sub bullet {
  # Draws a bullet point (filled circle) at the current text position
  #
  # Todo: need to remove the integer values here and replace with
  # percentages of the current line spacing to handle different fonts
  
  my $self    = shift;
  my $indent  = shift;
  my $bullet  = $self->{page}->gfx;
  my $x_coord = $self->{left_margin} + $self->{indent} + 4 + $indent - 20;
  my $y_coord = $self->{y_position} - 9 + ($self->print_flag ? 0 : $self->{line_spacing});
  my $radius  = 2;

  $bullet->circle($x_coord,$y_coord,$radius);
  $bullet->fillstroke;
}


#-----------------------------------------------------------------------

sub newline {
  my $self = shift;
  
  if ($self->print_flag) {
    $self->linefeed;
    $self->set_flag('newline');
  }
}


#-----------------------------------------------------------------------

sub linefeed {
  my $self = shift;
  
  $self->{y_position} -= $self->{line_spacing};
  $self->{x_position}  = $self->{left_margin} + $self->{indent};
    
  if ($self->{y_position} < ($self->{bottom_margin} + $self->{footer_height})) {
    my $style = $self->{style};
    $self->formfeed;
    $self->set_style($style);
  } 
}


#-----------------------------------------------------------------------

sub spacer {
  my $self = shift;
  $self->reset_space_flag;

  if ($self->print_flag) {
    $self->{y_position} -= $self->{spacer};
    $self->{x_position} = $self->{left_margin} + $self->{indent};

    if ($self->{y_position} < ($self->{bottom_margin} + $self->{footer_height})) {
      $self->formfeed;
    }
  }  
}


#-----------------------------------------------------------------------

sub formfeed {
  my $self = shift;
  $self->{page}       = $self->{pdf}->page;
  $self->{x_position} = $self->{left_margin} + $self->{indent};
  $self->{page_number}++;

  delete $self->{text};
  delete $self->{gfx};  
  $self->{gfx}  = $self->{page}->gfx;
  $self->{text} = $self->{page}->text;
  
  $self->{y_position}    = $self->{page_height} - $self->{top_margin} - $self->{line_spacing};
  $self->{header_height} = ($self->{header}) ? $self->generate_header : 0;
  $self->{footer_height} = ($self->{footer}) ? $self->generate_footer : 0;
  $self->{y_position}   -= $self->{header_height};

  $self->reset_print_flag;
}


#-----------------------------------------------------------------------
#-----------------------------------------------------------------------

# Page header and footer methods
#
# Future enhancement: pass the page number, filename, etc details as
# parameters to generate_header() and generate_footer(), allow these
# methods to be overridden by the user for custom page formatting.

#-----------------------------------------------------------------------

sub generate_header {
  my $self = shift;
  $self->set_style('header');
  
  my $header_padding = 2;
  my $header_spacing = 3;
  my $header_height  = $self->{text_size} + $header_spacing + $header_padding;
  
  # Draw header icon
  if ($self->{icon_img}) {
    my $icon_height_in_points = $self->{icon_height} * $self->{icon_scale};
    if ($icon_height_in_points > $self->{text_size}) {
      $header_height += ($icon_height_in_points - $self->{text_size});
    }
    my $ypos = $self->{page_height} - $self->{top_margin} - $icon_height_in_points;
    $self->{gfx}->image($self->{icon_img},$self->{left_margin},$ypos,$self->{icon_scale});
  }

  # Add page title
  my $x = $self->{page_width} - $self->{right_margin} - $self->{text}->advancewidth($self->{title});
  my $y = $self->{page_height} - $self->{top_margin} - $header_height + $header_spacing + $header_padding;
  $self->{text}->textlabel($x,$y,$self->{fontcache}->{$self->{font}},$self->{text_size},$self->{title});

  # Draw horizontal line
  $self->{gfx}->move($self->{left_margin},$self->{page_height}-$self->{top_margin}-$header_height + $header_padding);
  $self->{gfx}->line($self->{page_width}-$self->{right_margin},$self->{page_height}-$self->{top_margin}-$header_height + $header_padding);
  $self->{gfx}->stroke;
  
  return $header_height;
}


#-----------------------------------------------------------------------

sub generate_footer {
  my $self  = shift;
  $self->set_style('footer');

  # Add page footer
  my $t = 'Page '.$self->{page_number};
  my $x = $self->{page_width} - $self->{right_margin} - $self->{text}->advancewidth($t);
  my $y = $self->{bottom_margin};
  $self->{text}->textlabel($x,$y,$self->{fontcache}->{$self->{font}},$self->{text_size},$t);
  
  if ($self->{footer_text}) {
    $x = $self->{left_margin};
    $self->{text}->textlabel($x,$y,$self->{fontcache}->{$self->{font}},$self->{text_size},$self->{footer_text});
  }

  $self->{gfx} = $self->{page}->gfx unless (exists $self->{gfx});
  $self->{gfx}->move($self->{left_margin},$self->{bottom_margin}+10);
  $self->{gfx}->line($self->{page_width}-$self->{right_margin},$self->{bottom_margin}+10);
  $self->{gfx}->stroke;
  return 18;  # Footer height in points
}


#-----------------------------------------------------------------------

# Outline method
#
# On encountering a heading, generate_outline($text, $style, $page, $x, $y)
# is called, which use $style to determine depth in outline tree.
# $self->{outline_stack} is used to record the structure of PDF Outlines.

#-----------------------------------------------------------------------

sub generate_outline {
    my $self = shift;
    my $text = shift;
    $text =~ s/\s*$//; # remove trailing newlines and spaces

    my $style = shift;

    my $page  = shift;  #$self->{page_number};
    my $pdfpage = $self->{pdf}->openpage($page);

    my $x = shift;
    my $y = shift;

    my %level = (
        'head1' => 1,
        'head2' => 2,
        'head3' => 3,
        'head4' => 4,
    );
    my $level     = $level{$style};
    my $up_level  = $level - 1;
    my $cur_level = scalar @{$self->{outline_stack}} - 1;

    # finding the right place to insert new outline
    while ($cur_level < $up_level) {
        my $tail = $self->{outline_stack}[-1];
        push @{$self->{outline_stack}}, $tail->outline();
        $cur_level ++;
    }
    while ($cur_level > $up_level) {
        pop @{$self->{outline_stack}};
        $cur_level --;
    }
    my $tail = $self->{outline_stack}[-1];

    # create a new sub-outline, setting title and dest
    my $outline = $tail->outline();
    $outline->title($text);
    # $outline->dest($pdfpage);
    # $outline->dest($pdfpage, "-xyz" => [$self->{x_position}, $self->{y_position}, 0]);
    $outline->dest($pdfpage, "-xyz" => [$x, $y, 0]);

    push @{$self->{outline_stack}}, $outline;
}

#-----------------------------------------------------------------------
#-----------------------------------------------------------------------

# PDF file output
#
# When the PDF object goes out of scope, the generated PDF file will be
# printed to STDOUT. 
#
# Update - this doesn't work with PAR, need explicit $pdf->output() method

#-----------------------------------------------------------------------

sub output {
  my $self = shift;
  print $self->{pdf}->stringify;
  #$self->{pdf}->end;
}


#-----------------------------------------------------------------------
#-----------------------------------------------------------------------

# Text printing methods

#-----------------------------------------------------------------------

sub print {
  my $self = shift;
  my $text = shift;

  $self->newline;
  $self->print_text_with_style($text);
}


#-----------------------------------------------------------------------

sub print_text_with_style {
  my $self  = shift;
  my $text  = shift;
  my $style = shift;

  #warn "print_text_with_style called with style '$style', text '$text'\n";

  
  $self->set_style($style);
  
  # Remove double  spaces unless we are printing verbatim text
  unless ($self->{stylist}->{$self->{style}}->{verbatim}) {
    $text =~ s/(\s)\s+/$1/g;
  }

  if ($self->space_flag) {
    #
    # Note that this space appears in the default style,
    # but it should be printed in the previous style.
    #
    $self->reset_space_flag;
    $self->set_style('default');
    $self->print_word(' ');
    $self->set_style($style);
  }
  if ($text =~ s/\s+$//) {
    $self->set_space_flag;
  }

  while ($text =~ /(\s+|\S+)/g) {
    my $word = $1;
    $self->print_word($word);
  }
}


#-----------------------------------------------------------------------

sub print_word {
  my $self = shift;
  my $word = shift;
  
  # If we are at the start of a line (newline flag is set) and we are
  # NOT printing verbatim text, then suppress any whitespace.
  if ($self->flag('newline')) {
    #warn "newline flag set\n";
    #warn "x position = $self->{x_position}\n";
  }
  
  
  $self->set_print_flag;
  $self->clear_flag('newline');

    my $width = $self->{text}->advancewidth($word);
    if ($self->{x_position} + $width > $self->{page_width} - $self->{right_margin}) {
      
      # If the word will not fit on one line, split it up and recurse the 'print_word' sub
      if ($width > ($self->{page_width} - $self->{left_margin} - $self->{right_margin} - $self->{indent})) {
        my $fit = int(($self->{page_width} - $self->{left_margin} - $self->{right_margin} - $self->{indent}) / $self->{nspace});
        my @words = (substr($word,0,$fit),substr($word,$fit));
        #warn "Recursing... Word=$word Fit=$fit Xpos=$$self{x_position}\n";
        $self->print_word($_) foreach @words; 
        return;
      }
      
      $self->newline;
      if ($word =~ /^\s+$/) {
        unless ($self->{stylist}->{$self->{style}}->{verbatim}) {
          return;
        }
      }
    }
    $self->{x_position} += $self->{text}->textlabel($self->{x_position},
                                                    $self->{y_position},
                                                    $self->{fontcache}->{$self->{font}},
                                                    $self->{text_size},
                                                    $word,
                                                    -color => $self->{text_color});
    $self->set_print_flag;
    if ($self->{x_position} > $self->{page_width} - $self->{right_margin}) {
      $self->newline;
    }
}


#-----------------------------------------------------------------------
#-----------------------------------------------------------------------

# Text style methods

#-----------------------------------------------------------------------

sub set_style {
  my $self  = shift;
  my $style = shift || 'default';
  
  $style = (exists $self->{stylist}->{$style}) ? $style : 'default';
  #carp "Setting style to $style";
  
  # Create font object if necessary
  my $font = ($self->{stylist}->{$style}->{font} || $self->{font_face}) .
             ((exists $self->{stylist}->{$style}->{type}) ? '-'.$self->{stylist}->{$style}->{type} : '');
  unless (exists $self->{fontcache}->{$font}) {
    $self->{fontcache}->{$font} = $self->{pdf}->corefont($font);
  }
  
  $self->{style}      = $style;
  $self->{font}       = $font;
  $self->{text_color} = $self->{stylist}->{$style}->{color} || '#000000';
  $self->{text_size}  = $self->{stylist}->{$style}->{size}  || $self->{font_size};
  
  $self->{text}->font($self->{fontcache}->{$font},$self->{text_size});
  $self->{nspace}     = $self->{text}->advancewidth('n');
  $self->{mspace}     = $self->{text}->advancewidth('m');
}


#-----------------------------------------------------------------------
#-----------------------------------------------------------------------

1;

.\" Automatically generated by Pod::Man 2.25 (Pod::Simple 3.16)
.\"
.\" Standard preamble:
.\" ========================================================================
.de Sp \" Vertical space (when we can't use .PP)
.if t .sp .5v
.if n .sp
..
.de Vb \" Begin verbatim text
.ft CW
.nf
.ne \\$1
..
.de Ve \" End verbatim text
.ft R
.fi
..
.\" Set up some character translations and predefined strings.  \*(-- will
.\" give an unbreakable dash, \*(PI will give pi, \*(L" will give a left
.\" double quote, and \*(R" will give a right double quote.  \*(C+ will
.\" give a nicer C++.  Capital omega is used to do unbreakable dashes and
.\" therefore won't be available.  \*(C` and \*(C' expand to `' in nroff,
.\" nothing in troff, for use with C<>.
.tr \(*W-
.ds C+ C\v'-.1v'\h'-1p'\s-2+\h'-1p'+\s0\v'.1v'\h'-1p'
.ie n \{\
.    ds -- \(*W-
.    ds PI pi
.    if (\n(.H=4u)&(1m=24u) .ds -- \(*W\h'-12u'\(*W\h'-12u'-\" diablo 10 pitch
.    if (\n(.H=4u)&(1m=20u) .ds -- \(*W\h'-12u'\(*W\h'-8u'-\"  diablo 12 pitch
.    ds L" ""
.    ds R" ""
.    ds C` ""
.    ds C' ""
'br\}
.el\{\
.    ds -- \|\(em\|
.    ds PI \(*p
.    ds L" ``
.    ds R" ''
'br\}
.\"
.\" Escape single quotes in literal strings from groff's Unicode transform.
.ie \n(.g .ds Aq \(aq
.el       .ds Aq '
.\"
.\" If the F register is turned on, we'll generate index entries on stderr for
.\" titles (.TH), headers (.SH), subsections (.SS), items (.Ip), and index
.\" entries marked with X<> in POD.  Of course, you'll have to process the
.\" output yourself in some meaningful fashion.
.ie \nF \{\
.    de IX
.    tm Index:\\$1\t\\n%\t"\\$2"
..
.    nr % 0
.    rr F
.\}
.el \{\
.    de IX
..
.\}
.\"
.\" Accent mark definitions (@(#)ms.acc 1.5 88/02/08 SMI; from UCB 4.2).
.\" Fear.  Run.  Save yourself.  No user-serviceable parts.
.    \" fudge factors for nroff and troff
.if n \{\
.    ds #H 0
.    ds #V .8m
.    ds #F .3m
.    ds #[ \f1
.    ds #] \fP
.\}
.if t \{\
.    ds #H ((1u-(\\\\n(.fu%2u))*.13m)
.    ds #V .6m
.    ds #F 0
.    ds #[ \&
.    ds #] \&
.\}
.    \" simple accents for nroff and troff
.if n \{\
.    ds ' \&
.    ds ` \&
.    ds ^ \&
.    ds , \&
.    ds ~ ~
.    ds /
.\}
.if t \{\
.    ds ' \\k:\h'-(\\n(.wu*8/10-\*(#H)'\'\h"|\\n:u"
.    ds ` \\k:\h'-(\\n(.wu*8/10-\*(#H)'\`\h'|\\n:u'
.    ds ^ \\k:\h'-(\\n(.wu*10/11-\*(#H)'^\h'|\\n:u'
.    ds , \\k:\h'-(\\n(.wu*8/10)',\h'|\\n:u'
.    ds ~ \\k:\h'-(\\n(.wu-\*(#H-.1m)'~\h'|\\n:u'
.    ds / \\k:\h'-(\\n(.wu*8/10-\*(#H)'\z\(sl\h'|\\n:u'
.\}
.    \" troff and (daisy-wheel) nroff accents
.ds : \\k:\h'-(\\n(.wu*8/10-\*(#H+.1m+\*(#F)'\v'-\*(#V'\z.\h'.2m+\*(#F'.\h'|\\n:u'\v'\*(#V'
.ds 8 \h'\*(#H'\(*b\h'-\*(#H'
.ds o \\k:\h'-(\\n(.wu+\w'\(de'u-\*(#H)/2u'\v'-.3n'\*(#[\z\(de\v'.3n'\h'|\\n:u'\*(#]
.ds d- \h'\*(#H'\(pd\h'-\w'~'u'\v'-.25m'\f2\(hy\fP\v'.25m'\h'-\*(#H'
.ds D- D\\k:\h'-\w'D'u'\v'-.11m'\z\(hy\v'.11m'\h'|\\n:u'
.ds th \*(#[\v'.3m'\s+1I\s-1\v'-.3m'\h'-(\w'I'u*2/3)'\s-1o\s+1\*(#]
.ds Th \*(#[\s+2I\s-2\h'-\w'I'u*3/5'\v'-.3m'o\v'.3m'\*(#]
.ds ae a\h'-(\w'a'u*4/10)'e
.ds Ae A\h'-(\w'A'u*4/10)'E
.    \" corrections for vroff
.if v .ds ~ \\k:\h'-(\\n(.wu*9/10-\*(#H)'\s-2\u~\d\s+2\h'|\\n:u'
.if v .ds ^ \\k:\h'-(\\n(.wu*10/11-\*(#H)'\v'-.4m'^\v'.4m'\h'|\\n:u'
.    \" for low resolution devices (crt and lpr)
.if \n(.H>23 .if \n(.V>19 \
\{\
.    ds : e
.    ds 8 ss
.    ds o a
.    ds d- d\h'-1'\(ga
.    ds D- D\h'-1'\(hy
.    ds th \o'bp'
.    ds Th \o'LP'
.    ds ae ae
.    ds Ae AE
.\}
.rm #[ #] #H #V #F C
.\" ========================================================================
.\"
.IX Title "POD2PDF 1p"
.TH POD2PDF 1p "2012-04-20" "perl v5.14.2" "User Contributed Perl Documentation"
.\" For nroff, turn off justification.  Always turn off hyphenation; it makes
.\" way too many mistakes in technical documents.
.if n .ad l
.nh
.SH "NAME"
pod2pdf \- converts Pod to PDF format
.SH "DESCRIPTION"
.IX Header "DESCRIPTION"
pod2pdf converts documents written in Perl's \s-1POD\s0 (Plain Old Documentation)
format to \s-1PDF\s0 files.
.SS "Usage"
.IX Subsection "Usage"
.Vb 1
\& pod2pdf [options] input.pod >output.pdf
.Ve
.PP
If no input filename is specified, pod2pdf will read from \s-1STDIN\s0, e.g.
.PP
.Vb 1
\& perldoc \-u File::Find | pod2pdf [options] >File\-Find.pdf
.Ve
.SS "Options"
.IX Subsection "Options"
pod2pdf accepts the following command-line options:
.ie n .IP """\-\-output\-file""" 4
.el .IP "\f(CW\-\-output\-file\fR" 4
.IX Item "--output-file"
Sets the output filename for the generated \s-1PDF\s0 file. By default pod2pdf will output
to \s-1STDOUT\s0.
.ie n .IP """\-\-page\-size""" 4
.el .IP "\f(CW\-\-page\-size\fR" 4
.IX Item "--page-size"
Sets the page size to be used in the \s-1PDF\s0 file, can be set to any of the standard
paper sizes (A4, A5, Letter, etc). Defaults to A4.
.ie n .IP """\-\-page\-orientation""" 4
.el .IP "\f(CW\-\-page\-orientation\fR" 4
.IX Item "--page-orientation"
Controls if pages are produces in landscape or portrait format. Defaults to 'portrait'.
.ie n .IP """\-\-page\-width"", ""\-\-page\-height""" 4
.el .IP "\f(CW\-\-page\-width\fR, \f(CW\-\-page\-height\fR" 4
.IX Item "--page-width, --page-height"
Sets the width and height of the generated pages in points (for using non-standard
paper sizes).
.ie n .IP """\-\-left\-margin"", ""\-\-right\-margin"", ""\-\-top\-margin"", ""\-\-bottom\-margin""" 4
.el .IP "\f(CW\-\-left\-margin\fR, \f(CW\-\-right\-margin\fR, \f(CW\-\-top\-margin\fR, \f(CW\-\-bottom\-margin\fR" 4
.IX Item "--left-margin, --right-margin, --top-margin, --bottom-margin"
Allows each of the page margins (top, bottom, left, and right) to be individually
set in points.
.ie n .IP """\-\-margins""" 4
.el .IP "\f(CW\-\-margins\fR" 4
.IX Item "--margins"
Sets all page margins to the same size (specified in points).
.ie n .IP """\-\-header"", ""\-\-noheader""" 4
.el .IP "\f(CW\-\-header\fR, \f(CW\-\-noheader\fR" 4
.IX Item "--header, --noheader"
Controls if a header (containing the page title, and optional timestamp and icon)
will be included on each page. Defaults to on, so use \f(CW\*(C`\-\-noheader\*(C'\fR to disable.
.ie n .IP """\-\-title""" 4
.el .IP "\f(CW\-\-title\fR" 4
.IX Item "--title"
Sets the page title (defaults to the input filename).
.ie n .IP """\-\-timestamp""" 4
.el .IP "\f(CW\-\-timestamp\fR" 4
.IX Item "--timestamp"
Boolean option \- if set, includes the 'last modified' timestamp of the input file in
the page header.
.ie n .IP """\-\-icon""" 4
.el .IP "\f(CW\-\-icon\fR" 4
.IX Item "--icon"
Filename of an icon to be displayed in the top left corner of each page.
.ie n .IP """\-\-icon\-scale""" 4
.el .IP "\f(CW\-\-icon\-scale\fR" 4
.IX Item "--icon-scale"
Scaling value for the header icon (defaults to 0.25).
.ie n .IP """\-\-footer"", ""\-\-nofooter""" 4
.el .IP "\f(CW\-\-footer\fR, \f(CW\-\-nofooter\fR" 4
.IX Item "--footer, --nofooter"
Controls if a footer (containg the current page number and optional text string)
will be included on each page. By default the footer will be included, so use 
\&\f(CW\*(C`\-\-nofooter\*(C'\fR to disable.
.ie n .IP """\-\-footer\-text""" 4
.el .IP "\f(CW\-\-footer\-text\fR" 4
.IX Item "--footer-text"
Sets an optional footer text string that will be included in the bottom left corner
of each page.
.ie n .IP """\-\-outlines""" 4
.el .IP "\f(CW\-\-outlines\fR" 4
.IX Item "--outlines"
Adds outlines (bookmarks) to pdf according to headings (=head1, =head2, ...).
.ie n .IP """\-\-version""" 4
.el .IP "\f(CW\-\-version\fR" 4
.IX Item "--version"
Prints version number and exits.
.SS "Configuration files"
.IX Subsection "Configuration files"
Sets of command-line options may be saved into configuration files.
.PP
A configuration file contains options in the same format as used by pod2pdf on the command-line,
with one option given on each line of the file, e.g.
.PP
.Vb 2
\& \-\-page\-size A5
\& \-\-page\-orientation landscape
.Ve
.PP
To use a config file, invoke pod2pdf with the option \f(CW\*(C`@/path/to/configfile.conf\*(C'\fR.
.PP
For example, if you wanted to always include a company logo, timestamp, and copyright 
notice in your \s-1PDF\s0 files, create a file \fImycompany.conf\fR containing the following:
.PP
.Vb 3
\& \-\-icon "/path/to/your/logo.png"
\& \-\-footer\-text "Copyright 2007 MyCompany Limited"
\& \-\-timestamp
.Ve
.PP
Then invoke pod2pdf as:
.PP
.Vb 1
\& pod2pdf @/path/to/mycompany.conf input.pod >output.pdf
.Ve
.PP
If you create a config file called \fIpod2pdf.conf\fR and place this in the same 
directory as the pod2pdf script, it will be loaded as the default configuration.
.SH "POD ENTENSIONS"
.IX Header "POD ENTENSIONS"
As well as the standard \s-1POD\s0 commands (see perlpodspec), pod2pdf supports the following
extensions to the \s-1POD\s0 format:
.ie n .IP """=ff""" 4
.el .IP "\f(CW=ff\fR" 4
.IX Item "=ff"
The \f(CW\*(C`=ff\*(C'\fR command inserts a page bread (form feed) into the document.
.ie n .IP """O<...>""" 4
.el .IP "\f(CWO<...>\fR" 4
.IX Item "O<...>"
The \f(CW\*(C`O<...>\*(C'\fR formatting code inserts an external object (file) into the document. This
is primarily intended for embedding images, e.g.
.Sp
.Vb 1
\& O</path/to/figure1.jpg>
.Ve
.Sp
to insert diagrams, etc into documentation.
.Sp
pod2pdf supports the file types \s-1JPG\s0, \s-1GIF\s0, \s-1TIFF\s0, \s-1PNG\s0, and \s-1PNM\s0 for embedded objects.
.SH "DEPENDENCIES"
.IX Header "DEPENDENCIES"
pod2pdf requires the following modules to be installed:
.IP "\s-1PDF::API2\s0" 4
.IX Item "PDF::API2"
.PD 0
.IP "Pod::Escapes" 4
.IX Item "Pod::Escapes"
.IP "Getopt::ArgvFile" 4
.IX Item "Getopt::ArgvFile"
.PD
.PP
Additionally to use images, the modules File::Type and Image::Size must
be installed, and to specify alternative page sizes the Paper::Specs module
is required.
.SH "SEE ALSO"
.IX Header "SEE ALSO"
The pod2pdf homepage: <http://perl.jonallen.info/projects/pod2pdf>
.PP
For more information about \s-1POD\s0, read the perlpod manpage or see the \s-1POD\s0 page
on the Perl 5 Wiki (<http://www.perlfoundation.org/perl5/index.cgi?pod>).
.SH "COPYRIGHT and LICENSE"
.IX Header "COPYRIGHT and LICENSE"
Copyright (C) 2007 Jon Allen (\s-1JJ\s0) <jj@jonallen.info>
.PP
This software is licensed under the terms of the Artistic
License version 2.0.
.PP
For full license details, please read the file \fIartistic\-2_0.txt\fR 
included with this distribution, or see
http://www.perlfoundation.org/legal/licenses/artistic\-2_0.html <http://www.perlfoundation.org/legal/licenses/artistic-2_0.html>.

Changelog for pod2pdf (see http://perl.jonallen.info/projects/pod2pdf)

0.42 - 26 October 2007
  Fixed failing "compile.t" test

0.41 - 11 September 2007
  First public release to CPAN
  Now compatible with the latest PDF::API2 distribution

Format: http://www.debian.org/doc/packaging-manuals/copyright-format/1.0/
Upstream-Name: pod2pdf
Upstream-Contact: Jon Allen (JJ) <jj@jonallen.info>
Source: http://search.cpan.org/dist/pod2pdf/

Files: *
Copyright: 2007, Jon Allen (JJ) <jj@jonallen.info>
License: Artistic-2.0

Files: debian/*
Copyright: 2009-2010, Steffen Moeller <moeller@debian.org>
           2012, Guo Yixuan <culu.gyx@gmail.com>
License: Artistic-2.0 or GPL-3+

License: Artistic-2.0
    This program is free software; you can redistribute it and/or modify
    it under the terms of the Artistic License, which comes with Perl.
    On Debian GNU/Linux systems, the complete text of the Artistic License
    can be found in `/usr/share/common-licenses/Artistic'

License: GPL-3+
    This program is free software; you can redistribute it and/or modify
    it under the terms of the GNU General Public License as published by
    the Free Software Foundation; either version 3, or (at your option)
    any later version.
    On Debian GNU/Linux systems, the complete text of the GNU General
    Public License can be found in `/usr/share/common-licenses/GPL-3'

pod2pdf (0.42-4) unstable; urgency=low

  * Update debian/copyright.
  * Bump policy to 3.9.3 (no change needed).
  * Bump debhelper to 8.
  * Update tags in debian/patches.
  * Allow float point number in options: --page-width, --page-height, 
    --left-margin, --right-margin, --top-margin, --bottom-margin, 
    and --margins.
  * Fixed the overlapped text bug.
  * Add an option: --outlines (Closes: #651256).

 -- Guo Yixuan <culu.gyx@gmail.com>  Fri, 20 Apr 2012 11:09:18 +0800

pod2pdf (0.42-3) unstable; urgency=low

  * New Maintainer
  * Fixed PDF Producer info
  * Changed pod2pdf's manpage extension to .1p

 -- Guo Yixuan <culu.gyx@gmail.com>  Fri, 24 Feb 2012 17:07:43 +0800

pod2pdf (0.42-2) unstable; urgency=low

  * Documentation fixes (Closes: #541076).
  * Timestamp in ISO format (Closes: #550265)

 -- Steffen Moeller <moeller@debian.org>  Sun, 12 Sep 2010 17:25:16 +0200

pod2pdf (0.42-1) unstable; urgency=low

  * Initial Release (Closes: #536695).

 -- Steffen Moeller <moeller@debian.org>  Sun, 12 Jul 2009 15:49:19 +0200

pod2pdf for Debian
==================

The README file mentions Paper::Specs which is not yet available for
Debian. It would be lovely if someone could address the respective
packaging work.

dh-make-perl suggests the package to be named libpod2pdf-perl, but
since the package will mostly be used as a binary, I have decided
against that convention.

  Steffen Moeller <moeller@debian.org>  Sun, 12 Jul 2009 16:13:58 +0200

#!/usr/bin/perl

eval 'exec /usr/bin/perl  -S $0 ${1+"$@"}'
    if 0; # not running under some shell

# pod2pdf.pl - converts Pod to PDF format
#
# Copyright (C) 2007 Jon Allen <jj@jonallen.info>
#
# This software is licensed under the terms of the Artistic
# License version 2.0.
#
# For full license details, please read the file 'artistic-2_0.txt' 
# included with this distribution, or see
# http://www.perlfoundation.org/legal/licenses/artistic-2_0.html


#--Load required modules and activate Perl's safety features---------------

use strict;
use warnings;
use App::pod2pdf;
use File::Basename qw/basename/;
use File::Spec::Functions;
use FindBin qw/$Bin/;
use Getopt::ArgvFile qw/argvFile/;
use Getopt::Long;
use POSIX qw(locale_h);


#--Load config files-------------------------------------------------------

# Define config filename as <application_name>.conf
(my $configfile = basename($0)) =~ s/^(.*?)(?:\..*)?$/$1.conf/;

# Include config file from the same directory as the pod2pdf script
if (-e "$Bin/$configfile") {
  unshift @ARGV,'@'."$Bin/$configfile";
}

# If we have been packaged with PAR, include the config file from the
# application bundle
if ($ENV{PAR_TEMP} and -e "$ENV{PAR_TEMP}/inc/$configfile") {
  unshift @ARGV,'@'."$ENV{PAR_TEMP}/inc/$configfile";
}

argvFile();  # Process @ARGV to load specified config files. (Function
             # from Getopt::ArgvFile - interpolates "@filename" entries
             # in @ARGV with the contents of the specified file)
             
             

#--------------------------------------------------------------------------

sub dateISO8601 ($) {
    my ($timeval) = shift || time;

    warn "time: $timeval";

    my ($yyyy, $MM, $dd, $hh, $mm) = (localtime $timeval)[5, 4, 3, 2, 1];
    $yyyy += 1900;
    $MM++;

    sprintf "$yyyy-%02d-%02d %02d:%02d", $MM, $dd, $hh, $mm;
}

#--Parse command-line options----------------------------------------------

my %options = (
  'page-height'      => '=f',
  'page-width'       => '=f',
  'page-size'        => '=s',
  'page-orientation' => '=s',
  'margins'          => '=f',
  'left-margin'      => '=f',
  'right-margin'     => '=f',
  'top-margin'       => '=f',
  'bottom-margin'    => '=f',
  'header'           => '!',
  'footer'           => '!',
  'title'            => '=s',
  'footer-text'      => '=s',
  'icon'             => '=s',
  'icon-scale'       => '=s',
  'timestamp'        => '!',
  'output-file'      => '=s',
  'outlines'         => '!',
);

my %config;
GetOptions(\%config,
  optionspec(%options),
  version => sub{ print "This is pod2pdf, version $App::pod2pdf::VERSION\n"; exit }
) or die("[Error] Could not parse options");

my $file = $ARGV[0];
my $time = $file ? $^T - (-M $file) * 24 * 60 * 60 : time;

$config{title}  = $file ? $file : 'STDIN' unless exists $config{title};
$config{title} .= ' - ' .  dateISO8601 $time if exists $config{timestamp};

#--Set output location-----------------------------------------------------

if (my $outfile = $config{output_file}) {
  open STDOUT,'>',$outfile or die("Cannot open output file $outfile: $!\n")
}


#--Tell the OS we are going to create binary data--------------------------

setlocale(LC_ALL,'C');
binmode *STDOUT;


#--Parse our Pod-----------------------------------------------------------

my $parser = App::pod2pdf->new(%config);
(@ARGV) ? $parser->parse_from_file($ARGV[0]) : $parser->parse_from_filehandle(\*STDIN);

$parser->output;

exit;


#--------------------------------------------------------------------------

sub optionspec {
  my %option_specs = @_;
  my @getopt_list;
  while (my ($option_name,$spec) = each %option_specs) {
    (my $variable_name = $option_name) =~ tr/-/_/;
    (my $nospace_name  = $option_name) =~ s/-//g;
    my $getopt_name = ($variable_name ne $option_name) ? "$variable_name|$option_name|$nospace_name" : $option_name;
    push @getopt_list,"$getopt_name$spec";
  }
  return @getopt_list;
}


#--------------------------------------------------------------------------
#--------------------------------------------------------------------------

=head1 NAME

pod2pdf - converts Pod to PDF format

=head1 DESCRIPTION

pod2pdf converts documents written in Perl's POD (Plain Old Documentation)
format to PDF files.

=head2 Usage

 pod2pdf [options] input.pod >output.pdf

If no input filename is specified, pod2pdf will read from STDIN, e.g.

 perldoc -u File::Find | pod2pdf [options] >File-Find.pdf

=head2 Options

pod2pdf accepts the following command-line options:

=over

=item C<--output-file>

Sets the output filename for the generated PDF file. By default pod2pdf will output
to STDOUT.

=item C<--page-size>

Sets the page size to be used in the PDF file, can be set to any of the standard
paper sizes (A4, A5, Letter, etc). Defaults to A4.

=item C<--page-orientation>

Controls if pages are produces in landscape or portrait format. Defaults to 'portrait'.

=item C<--page-width>, C<--page-height>

Sets the width and height of the generated pages in points (for using non-standard
paper sizes).

=item C<--left-margin>, C<--right-margin>, C<--top-margin>, C<--bottom-margin>

Allows each of the page margins (top, bottom, left, and right) to be individually
set in points. 

=item C<--margins>

Sets all page margins to the same size (specified in points).

=item C<--header>, C<--noheader>

Controls if a header (containing the page title, and optional timestamp and icon)
will be included on each page. Defaults to on, so use C<--noheader> to disable.

=item C<--title>

Sets the page title (defaults to the input filename).

=item C<--timestamp>

Boolean option - if set, includes the 'last modified' timestamp of the input file in
the page header.

=item C<--icon>

Filename of an icon to be displayed in the top left corner of each page.

=item C<--icon-scale>

Scaling value for the header icon (defaults to 0.25).

=item C<--footer>, C<--nofooter>

Controls if a footer (containg the current page number and optional text string)
will be included on each page. By default the footer will be included, so use 
C<--nofooter> to disable.

=item C<--footer-text>

Sets an optional footer text string that will be included in the bottom left corner
of each page.

=item C<--outlines>

Adds outlines (bookmarks) to pdf according to headings (=head1, =head2, ...).

=item C<--version>

Prints version number and exits.

=back

=head2 Configuration files

Sets of command-line options may be saved into configuration files. 

A configuration file contains options in the same format as used by pod2pdf on the command-line,
with one option given on each line of the file, e.g.

 --page-size A5
 --page-orientation landscape

To use a config file, invoke pod2pdf with the option C<@/path/to/configfile.conf>.

For example, if you wanted to always include a company logo, timestamp, and copyright 
notice in your PDF files, create a file F<mycompany.conf> containing the following:

 --icon "/path/to/your/logo.png"
 --footer-text "Copyright 2007 MyCompany Limited"
 --timestamp

Then invoke pod2pdf as:

 pod2pdf @/path/to/mycompany.conf input.pod >output.pdf

If you create a config file called F<pod2pdf.conf> and place this in the same 
directory as the pod2pdf script, it will be loaded as the default configuration.

=head1 POD ENTENSIONS

As well as the standard POD commands (see L<perlpodspec>), pod2pdf supports the following
extensions to the POD format:

=over

=item C<=ff>

The C<=ff> command inserts a page bread (form feed) into the document.

=item C<< OE<lt>...E<gt> >>

The C<< OE<lt>...E<gt> >> formatting code inserts an external object (file) into the document. This
is primarily intended for embedding images, e.g.

 O</path/to/figure1.jpg>

to insert diagrams, etc into documentation.

pod2pdf supports the file types JPG, GIF, TIFF, PNG, and PNM for embedded objects.

=back

=head1 DEPENDENCIES

pod2pdf requires the following modules to be installed:

=over

=item L<PDF::API2>

=item L<Pod::Escapes>

=item L<Getopt::ArgvFile>

=back

Additionally to use images, the modules L<File::Type> and L<Image::Size> must
be installed, and to specify alternative page sizes the L<Paper::Specs> module
is required.

=head1 SEE ALSO

The pod2pdf homepage: L<http://perl.jonallen.info/projects/pod2pdf>

For more information about POD, read the L<perlpod> manpage or see the POD page
on the Perl 5 Wiki (L<http://www.perlfoundation.org/perl5/index.cgi?pod>).

=head1 COPYRIGHT and LICENSE

Copyright (C) 2007 Jon Allen (JJ) <jj@jonallen.info>

This software is licensed under the terms of the Artistic
License version 2.0.

For full license details, please read the file F<artistic-2_0.txt> 
included with this distribution, or see
L<http://www.perlfoundation.org/legal/licenses/artistic-2_0.html>.










